﻿Shader "RenderingPractice/Normal Map In Tangent Space"//切线空间下计算法线贴图
{
    Properties
    {
		_Color("Color Tint",Color) = (1,1,1,1)//颜色基调
        _MainTex ("Main Texture", 2D) = "white" {}//主纹理
		_BumpMap("Normal Map",2D) = "bump"{}//法线贴图
		_BumpScale("Bump Scale(法线强度)",Range(-1.5,1.5)) = 0.0//法线强度
		_Specular("Specular Color",Color) = (1,1,1,1)//镜面反射颜色
		_Gloss("Gloss(光泽度)",Range(8.0,256.0)) = 8.0//光泽度（表面光滑度，影响高光大小）
    }
    SubShader
    {
        Tags { "RenderType"="Opaque" "LightMode" = "ForwardBase" }
        LOD 100

        Pass
        {
            CGPROGRAM
            #pragma vertex VertexFunction
            #pragma fragment FragmentFunction

            #include "UnityCG.cginc"
			#include "Lighting.cginc"

			fixed4 _Color;//颜色基调
			sampler2D _MainTex;//主纹理（漫反射颜色）
            float4 _MainTex_ST;//主纹理 缩放/平移
			sampler2D _BumpMap;//法线贴图
			float4 _BumpMap_ST;
			float _BumpScale;//法线强度
			fixed4 _Specular;//镜面反射颜色
			float _Gloss;//光泽度（表面光滑度，影响高光大小）

            struct a2v
            {
                float4 vertex : POSITION;
				float3 normal : NORMAL;
				float4 tangent : TANGENT;//顶点切线（w分量决定与法线叉乘后向量的方向（即 副切线的方向））
                float4 texcoord : TEXCOORD0;//第一组纹理的UV
            };

            struct v2f
            {
				float4 vertex : SV_POSITION;
				float3 lightDir : TEXCOORD0;
				float3 viewDir : TEXCOORD1;
                float4 uv : TEXCOORD2;//第一组纹理的UV
            };

            v2f VertexFunction (a2v v)
            {
                v2f o;
                o.vertex = UnityObjectToClipPos(v.vertex);
				//计算 模型空间=>切线空间 的变换矩阵：
				float3 bitangent = cross( normalize(v.tangent.xyz) , normalize(v.normal)) * v.tangent.w;//副切线
				float3x3 rotation = float3x3(v.tangent.xyz , bitangent , v.normal);
				//TANGENT_SPACE_ROTATION;//在 UnityCG 中计算得到 模型空间=>切线空间 的变换矩阵 rotation
				//切线空间下的光线方向、视角方向
				o.lightDir = mul(rotation , ObjSpaceLightDir(v.vertex)).xyz;
				o.viewDir = mul(rotation , ObjSpaceViewDir(v.vertex)).xyz;
				//纹理坐标变换：
                o.uv.xy = TRANSFORM_TEX(v.texcoord , _MainTex);
				o.uv.zw = TRANSFORM_TEX(v.texcoord , _BumpMap);
                return o;
            }

            fixed4 FragmentFunction (v2f i) : SV_Target
            {
				fixed3 tangentLightDir = normalize(i.lightDir);//l
				fixed3 tangentViewDir = normalize(i.viewDir);//v
				//1.法线贴图采样：
				fixed4 packedNormal = tex2D(_BumpMap , i.uv.zw);
				fixed3 tangentNormalDir = UnpackNormal(packedNormal);
				tangentNormalDir.xy *= _BumpScale;//n
				//2.纹理采样，计算 albedo：
				fixed3 albedo = tex2D(_MainTex , i.uv.xy).rgb * _Color.rgb;
				//1.环境光：
				fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz * albedo;
				//2.漫反射：
				fixed3 diffuseColor = _LightColor0.rgb * albedo.rgb * saturate(dot(tangentNormalDir , tangentLightDir)) * albedo;
				//3.镜面反射：（Blinn-Phong）
				float3 hDir = normalize(tangentViewDir + tangentLightDir);//h
				fixed3 specularColor = _LightColor0.rgb * _Specular.rgb * pow(saturate(dot(tangentNormalDir , hDir)) , _Gloss);
				//4.光照颜色：
				fixed4 c = fixed4(ambient + diffuseColor + specularColor , 1.0);
                return c;
            }
            ENDCG
        }
    }
	FallBack "Specular"
}
